NAG Library Routine Document

F03AFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F03AFF computes an LU factorization of a real matrix, with partial pivoting, and evaluates the determinant.

2 Specification

```
SUBROUTINE F03AFF(N, EPS, A, LDA, D1, ID, P, IFAIL)
INTEGER           N, LDA, ID, IFAIL
double precision  EPS, A(LDA,*), D1, P(N)
```

3 Description

F03AFF computes an LU factorization of a real matrix A with partial pivoting: $PA = LU$, where P is a permutation matrix, L is lower triangular and U is unit upper triangular. The determinant of A is the product of the diagonal elements of L with the correct sign determined by the row interchanges.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Parameters

1: N – INTEGER *Input*

On entry: the dimension of the array P as declared in the (sub)program from which F03AFF is called. n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: EPS – ***double precision*** *Input*

On entry: is no longer required by F03AFF but is retained for backwards compatibility.

3: A(LDA,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the n by n matrix A.

On exit: A is overwritten by the lower triangular matrix L and the off-diagonal elements of the upper triangular matrix U. The unit diagonal elements of U are not stored.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F03AFF is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

5: D1 – ***double precision*** *Output*
6: ID – INTEGER *Output*

On exit: the determinant of A is given by D1 $\times$ 2.0^{ID}. It is given in this form to avoid overflow or underflow.

7: P(N) – ***double precision*** array *Output*

On exit: P(i) gives the row index of the ith pivot.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

The matrix A is singular, possibly due to rounding errors. The factorization could not be completed. D1 and ID are set to zero.

IFAIL $= 2$

On entry, N < 0,
or LDA $<$ max(1, N).

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis, see page 107 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F03AFF is approximately proportional to n^3.

9 Example

This example computes the LU factorization with partial pivoting, and calculates the determinant, of the real matrix

$$\begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix}.$$

9.1 Program Text

```
*     F03AFF Example Program Text
*     Mark 15 Revised. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NMAX, LDA
      DOUBLE PRECISION TWO
      PARAMETER        (NMAX=8,LDA=NMAX,TWO=2.0D0)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION D1, EPS
      INTEGER          I, ID, IFAIL, J, N
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), P(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION X02AJF
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         F03AFF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F03AFF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GE.0 .AND. N.LE.NMAX) THEN
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
         EPS = X02AJF()
         IFAIL = 1
*
         CALL F03AFF(N,EPS,A,LDA,D1,ID,P,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,*) 'Array A after factorization'
            DO 20 I = 1, N
               WRITE (NOUT,99998) (A(I,J),J=1,N)
   20       CONTINUE
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Array P'
            WRITE (NOUT,99998) (P(I),I=1,N)
            WRITE (NOUT,*)
            WRITE (NOUT,99997) 'D1 = ', D1, ' ID = ', ID
            D1 = D1*TWO**ID
            WRITE (NOUT,*)
            WRITE (NOUT,99997) 'Value of determinant = ', D1
         ELSE
            WRITE (NOUT,99996) IFAIL
         END IF
      ELSE
         WRITE (NOUT,99999) 'N is out of range: N = ', N
      END IF
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,8F9.4)
99997 FORMAT (1X,A,F9.4,A,I2)
99996 FORMAT (1X,/1X,' ** F03AFF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
F03AFF Example Program Data
  3
   33   16   72
  -24  -10  -57
   -8   -4  -17
```

9.3 Program Results

```
 F03AFF Example Program Results

 Array A after factorization
   -8.0000   0.5000   2.1250
  -24.0000   2.0000  -3.0000
   33.0000  -0.5000   0.3750

 Array P
    3.0000   2.0000   3.0000

 D1 =    0.3750 ID =  4

 Value of determinant =    6.0000
```